


Semantic Web Information Management

Norberto Fernández

Telematics Engineering Department

berto@it.uc3m.es



Motivation

- n Module 1: An ontology models a domain of knowledge...
 - n Module 2: ... using the ontology as vocabulary, we represent (meta)data in RDF...
 - n ..., OK, so we have tons of triples representing all our information: What do we do with them?
 - n RDF storage
 - n RDF querying
- Semantic Web Information management

Introduction

- n Module 3: Semantic Web information management
 - n RDF querying
 - n We will see this in theoretical lessons
 - n SPARQL: <http://www.w3.org/TR/rdf-sparql-query/>
 - n RDF storage
 - n We will see this mainly in lab
 - n Jena: <http://jena.sourceforge.net/>

RDF Querying

Query Languages for RDF

- n Several proposals out there...
 - n RDQL, Triple, SeRQL, RQL, RDFQL, N3, etc.
 - n <http://www.aifb.uni-karlsruhe.de/WBS/pha/rdf-query/>
- n But nowadays there is also a W3C recommendation: SPARQL

SPARQL

- n Pronounced “*sparkle*”
- n Formerly an acronym:
 - n *SPARQL Protocol And RDF Query Language*
- n Defined by the RDF Data Access Working Group of the W3C
- n Three related specifications:
 - n SPARQL Query Language for RDF
 - n Query language
 - n SPARQL Protocol for RDF
 - n Protocol to access SPARQL endpoints
 - n SPARQL Query Results XML Format
 - n Representing query results in XML
- n W3C recommendations since January 2008

SPARQL Syntax: IRIs (I)

- n Subset of URIs used in SPARQL queries as resources references
- n RDF URI references containing "<", ">", double quote, space, "{", "}", "|", "\", "^", and "~" are not IRIs
- n The behavior of a SPARQL query against RDF statements composed of such RDF URI references is not defined

SPARQL Syntax: IRIs (II)

- n Three mechanisms to represents IRIs:

```
<http://example.org/book/book1>
```

Absolute IRI

```
BASE <http://example.org/book/>  
<book1>
```

Relative IRI

```
PREFIX book: <http://example.org/book/>  
book:book1
```

Prefixed name

SPARQL Syntax: Variables

n Most forms of SPARQL query contain a set of triple patterns called a *basic graph pattern*. Triple patterns are like RDF triples except that each of the subject, predicate and object may be a variable

n Use ? Or \$ to denote variables

n Example:

```
<http://example.org/book/book1> <http://purl.org/dc/elements/1.1/title> ?title .
```



Basic graph pattern (in this case, simply a triple pattern)

SPARQL Syntax: Literals

- n Sometimes literals are included as part of SPARQL queries
- n Literal syntax:
 - n "chat" (a string)
 - n 'chat'@fr (a string with language tag "fr")
 - n "xyz"^^http://example.org/ns/userDatatype (user datatype)
 - n "abc"^^appNS:appDataType (another user datatype)
 - n 1, which is the same as "1"^^xsd:integer
 - n 1.300, which is the same as "1.300"^^xsd:decimal
 - n 1.0e6, which is the same as "1.0e6"^^xsd:double
 - n true, which is the same as "true"^^xsd:boolean

SPARQL Syntax: Hello World query

Data

```
<http://example.org/book/book1> <http://purl.org/dc/elements/1.1/title> "SPARQL Tutorial" .
```

SPARQL Query

```
SELECT ?title
WHERE
{
  <http://example.org/book/book1> <http://purl.org/dc/elements/1.1/title> ?title .
}
```

Basic graph pattern



Results

title
"SPARQL Tutorial"

Source: <http://www.w3.org/TR/rdf-sparql-query/>

SPARQL: Another example

RDF Data
(N3 format)

```
@prefix foaf: <http://xmlns.com/foaf/0.1/> .  
_:a foaf:name "Johnny Lee Outlaw" .  
_:a foaf:mbox <mailto:jlow@example.com> .  
_:b foaf:name "Peter Goodguy" .  
_:b foaf:mbox <mailto:peter@example.org> .  
_:c foaf:mbox <mailto:carol@example.org> .
```

```
PREFIX foaf: <http://xmlns.com/foaf/0.1/>  
SELECT ?name ?mbox  
WHERE  
{ ?x foaf:name ?name .  
  ?x foaf:mbox ?mbox }
```

SPARQL query

Results

name	mbox
"Johnny Lee Outlaw"	<mailto:jlow@example.com>
"Peter Goodguy"	<mailto:peter@example.org>

Source: <http://www.w3.org/TR/rdf-sparql-query/>

SPARQL: Queries with literals

```
@prefix dt:    <http://example.org/datatype#> .
@prefix ns:    <http://example.org/ns#> .
@prefix :      <http://example.org/ns#> .
@prefix xsd:   <http://www.w3.org/2001/XMLSchema#> .

:x  ns:p      "cat"@en .
:y  ns:p      "42"^^xsd:integer .
:z  ns:p      "abc"^^dt:specialDatatype .
```

```
SELECT ?v WHERE { ?v ?p "cat"@en }
```

v
<http://example.org/ns#x>

```
SELECT ?v WHERE { ?v ?p 42 }
```

v
<http://example.org/ns#y>

```
SELECT ?v WHERE { ?v ?p "abc"^^<http://example.org/datatype#specialDatatype> }
```

v
<http://example.org/ns#z>

Source: <http://www.w3.org/TR/rdf-sparql-query/>

SPARQL: Abbreviations

n Predicate-object lists

- n For triples with the same subject

```
?x foaf:name ?name ;  
   foaf:mbox ?mbox .
```

n Object lists

- n For triples with the same subject and predicate

```
?x foaf:nick "Alice" , "Alice_" .
```

n Abbreviation for rdf:type a

```
?x a :Class1 .
```

SPARQL: Restrictions (Filters)

- n SPARQL FILTERs restrict solutions to those for which the filter expression evaluates to TRUE
- n Filters can use functions (like regex or lang) and operators:
 - n =, !=, <, <=, >=, >, &&, ||
- n Restriction on solutions over the whole group of graph patterns in which the filter appears
 - n The order of the filter in the group does not matter

SPARQL: Filters (I)

```
@prefix foaf:      <http://xmlns.com/foaf/0.1/> .

_:a  foaf:name     "Alice".
_:a  foaf:mbox     <mailto:alice@work.example> .

_:b  foaf:name     "Ms A.".
_:b  foaf:mbox     <mailto:alice@work.example> .
```

```
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
SELECT ?name1 ?name2
WHERE { ?x foaf:name ?name1 ;
        foaf:mbox ?mbox1 .
        ?y foaf:name ?name2 ;
        foaf:mbox ?mbox2 .
        FILTER (?mbox1 = ?mbox2 && ?name1 != ?name2)
}
```

name1	name2
"Alice"	"Ms A."
"Ms A."	"Alice"

Source: <http://www.w3.org/TR/rdf-sparql-query/>

SPARQL: Filters (II)

- n Filtering using regular expressions (regex)
 - n Works for string literals without language tag. In order to be used with other literals function `str`

```
@prefix foaf:      <http://xmlns.com/foaf/0.1/> .  
  
_:a foaf:name      "Alice".  
_:a foaf:mbox      <mailto:alice@work.example> .  
  
_:b foaf:name      "Bob" .  
_:b foaf:mbox      <mailto:bob@home.example> .
```

```
PREFIX foaf: <http://xmlns.com/foaf/0.1/>  
SELECT ?name ?mbox  
WHERE { ?x foaf:name ?name ;  
        foaf:mbox ?mbox .  
        FILTER regex(str(?mbox), "@work.example") }
```

name	mbox
"Alice"	<mailto:alice@work.example>

SPARQL: Filters (III)

```
@prefix foaf:      <http://xmlns.com/foaf/0.1/> .  
  
_:a foaf:name      "Robert"@EN.  
_:a foaf:name      "Roberto"@ES.  
_:a foaf:mbox      <mailto:bob@work.example> .
```

```
PREFIX foaf: <http://xmlns.com/foaf/0.1/>  
SELECT ?name ?mbox  
WHERE { ?x foaf:name ?name ;  
        foaf:mbox ?mbox .  
        FILTER ( lang(?name) = "ES" ) }
```

name	mbox
"Roberto"@ES	<mailto:bob@work.example>

Source: <http://www.w3.org/TR/rdf-sparql-query/>

SPARQL: Optional (I)

- n Regular, complete structures cannot be assumed in all RDF graphs
- n It is useful to be able to have queries that allow information to be added to the solution where the information is available, but do not reject the solution because some part of the query pattern does not match Optional matching
- n A graph pattern may have zero or more optional graph patterns

SPARQL: Optional (II)

```
@prefix foaf:      <http://xmlns.com/foaf/0.1/> .
@prefix rdf:      <http://www.w3.org/1999/02/22-rdf-syntax-ns#> .

_:a  rdf:type      foaf:Person .
_:a  foaf:name     "Alice" .
_:a  foaf:mbox     <mailto:alice@example.com> .
_:a  foaf:mbox     <mailto:alice@work.example> .

_:b  rdf:type      foaf:Person .
_:b  foaf:name     "Bob" .
```

```
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
SELECT ?name ?mbox
WHERE { ?x foaf:name ?name .
       OPTIONAL { ?x foaf:mbox ?mbox }
}
```

name	mbox
"Alice"	<mailto:alice@example.com>
"Alice"	<mailto:alice@work.example>
"Bob"	

Source: <http://www.w3.org/TR/rdf-sparql-query/>

SPARQL: Union (I)

- n SPARQL provides a means of combining graph patterns so that one of several alternative graph patterns may match
- n If more than one of the alternatives matches, all the possible pattern solutions are found
- n Alternative graph patterns are represented with the keyword union

SPARQL: Union (II)

```
@prefix dc10: <http://purl.org/dc/elements/1.0/> .
@prefix dc11: <http://purl.org/dc/elements/1.1/> .

_:a dc10:title      "SPARQL Query Language Tutorial" .
_:a dc10:creator    "Alice" .

_:b dc11:title      "SPARQL Protocol Tutorial" .
_:b dc11:creator    "Bob" .

_:c dc10:title      "SPARQL" .
_:c dc11:title      "SPARQL (updated)" .
```

```
PREFIX dc10: <http://purl.org/dc/elements/1.0/>
PREFIX dc11: <http://purl.org/dc/elements/1.1/>

SELECT ?title
WHERE { { ?book dc10:title ?title } UNION { ?book dc11:title ?title } }
```

title
"SPARQL Protocol Tutorial"
"SPARQL"
"SPARQL (updated)"
"SPARQL Query Language Tutorial"

Source:

<http://www.w3.org/TR/rdf-sparql-query/>

Specifying the RDF Dataset (I)

- n Many RDF data stores hold multiple RDF graphs and record information about each graph, allowing an application to make queries that involve information from more than one graph
- n A SPARQL query may specify the dataset to be used for matching by using the FROM clause and the FROM NAMED clause

Specifying the RDF Dataset (II)

n FROM

- n An optional clause that provides the URI of the dataset to use, pointing to a local file or a URL of a graph somewhere on the Web

n FROM NAMED

- n Loads a graph and gives it a name (an IRI)
- n This name can be accessed from queries using GRAPH

Specifying the RDF Dataset (III)

```
# Default graph (stored at http://example.org/dft.ttl)
@prefix dc: <http://purl.org/dc/elements/1.1/> .

<http://example.org/bob>    dc:publisher  "Bob Hacker" .
<http://example.org/alice> dc:publisher  "Alice Hacker" .
```

```
# Named graph: http://example.org/bob
@prefix foaf: <http://xmlns.com/foaf/0.1/> .

_:a foaf:name "Bob" .
_:a foaf:mbox <mailto:bob@oldcorp.example.org> .
```

```
# Named graph: http://example.org/alice
@prefix foaf: <http://xmlns.com/foaf/0.1/> .

_:a foaf:name "Alice" .
_:a foaf:mbox <mailto:alice@work.example.org> .
```

```
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
PREFIX dc: <http://purl.org/dc/elements/1.1/>

SELECT ?who ?g ?mbox
FROM <http://example.org/dft.ttl>
FROM NAMED <http://example.org/alice>
FROM NAMED <http://example.org/bob>
WHERE
{
  ?g dc:publisher ?who .
  GRAPH ?g { ?x foaf:mbox ?mbox }
}
```

Source: <http://www.w3.org/TR/rdf-sparql-query/>

SPARQL: Order By

- n Establishes the order of a solution sequence
 - n ASC() or DESC()

```
PREFIX      :      <http://example.org/ns#>
PREFIX foaf:  <http://xmlns.com/foaf/0.1/>
PREFIX xsd:   <http://www.w3.org/2001/XMLSchema#>

SELECT ?name
WHERE { ?x foaf:name ?name ; :empId ?emp }
ORDER BY DESC(?emp)
```

Source: <http://www.w3.org/TR/rdf-sparql-query/>

SPARQL: Distinct

n Eliminates duplicate solutions

```
@prefix foaf: <http://xmlns.com/foaf/0.1/> .

_:x foaf:name "Alice" .
_:x foaf:mbox <mailto:alice@example.com> .

_:y foaf:name "Alice" .
_:y foaf:mbox <mailto:asmith@example.com> .

_:z foaf:name "Alice" .
_:z foaf:mbox <mailto:alice.smith@example.com> .
```

```
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
SELECT DISTINCT ?name WHERE { ?x foaf:name ?name }
```

name
"Alice"

Source: <http://www.w3.org/TR/rdf-sparql-query/>

SPARQL: Limit

- n Puts an upper bound on the number of solutions returned
- n A LIMIT of 0 would cause no results to be returned. A limit may not be negative

```
PREFIX foaf:    <http://xmlns.com/foaf/0.1/>

SELECT ?name
WHERE { ?x foaf:name ?name }
LIMIT 20
```

Source: <http://www.w3.org/TR/rdf-sparql-query/>

SPARQL: Offset

- n Causes the solutions generated to start after the specified number of solutions
- n An OFFSET of zero has no effect
- n Using LIMIT and OFFSET requires the order to be made predictable by using ORDER BY

```
PREFIX foaf:      <http://xmlns.com/foaf/0.1/>

SELECT  ?name
WHERE   { ?x foaf:name ?name }
ORDER BY ?name
LIMIT  5
OFFSET 10
```

Source: <http://www.w3.org/TR/rdf-sparql-query/>

SPARQL: Construct (I)

n CONSTRUCT

- n The CONSTRUCT query form returns a single RDF graph specified by a graph template. The result is an RDF graph formed by taking each query solution in the solution sequence, substituting for the variables in the graph template, and combining the triples into a single RDF graph by set union

SPARQL: Construct (II)

```
@prefix foaf: <http://xmlns.com/foaf/0.1/> .  
  
_:a foaf:name "Alice" .  
_:a foaf:mbox <mailto:alice@example.org> .
```

```
PREFIX foaf: <http://xmlns.com/foaf/0.1/>  
PREFIX vcard: <http://www.w3.org/2001/vcard-rdf/3.0#>  
CONSTRUCT { <http://example.org/person#Alice> vcard:FN ?name }  
WHERE { ?x foaf:name ?name }
```

```
@prefix vcard: <http://www.w3.org/2001/vcard-rdf/3.0#> .  
  
<http://example.org/person#Alice> vcard:FN "Alice" .
```

Source: <http://www.w3.org/TR/rdf-sparql-query/>

SPARQL: Ask (I)

n ASK

- n Applications can use the ASK query form to test whether or not a query pattern has a solution. No information is returned about the possible query solutions, just whether or not a solution exists
- n The result is thus “yes” / “no”

SPARQL: Ask (II)

```
@prefix foaf:      <http://xmlns.com/foaf/0.1/> .  
  
_:a foaf:name      "Alice" .  
_:a foaf:homepage  <http://work.example.org/alice/> .  
  
_:b foaf:name      "Bob" .  
_:b foaf:mbox      <mailto:bob@work.example> .
```

```
PREFIX foaf:      <http://xmlns.com/foaf/0.1/>  
ASK { ?x foaf:name "Alice" }
```

Source: <http://www.w3.org/TR/rdf-sparql-query/>

SPARQL Protocol for RDF

- n Defines the protocol for issuing SPARQL queries over remote repositories and receiving the results
 - n Defined using WSDL (as a Web Service)
 - n The service contains one interface, SparqlQuery, which contains one operation, query
 - n Bindings to HTTP and to SOAP are also defined

SPARQL Protocol for RDF

- n The query operation...
 - n ... receives a query-request message that contains the query in SPARQL (a string) and information about the RDF datasets where the query is performed (default-graph-uri, named-graph-uri)
 - n ... returns a query-result message that contains..
 - n SELECT and ASK queries a result set represented as an SPARQL results document in XML format (see next)
 - n CONSTRUCT queries an RDF document
 - n Faults are also defined
 - n malformed-query, query-request-refused

SPARQL Query Results XML Format

- n Defines an XML document format for representing the results of SPARQL SELECT and ASK queries
 - n Three parts:
 - n Sparql element
 - n Header
 - n Results

```
<sparql xmlns="http://www.w3.org/2005/sparql-results#">

  <head>
    <variable name="x"/>
    <variable name="hpage"/>
    <variable name="name"/>
    <variable name="age"/>
    <variable name="mbox"/>
    <variable name="friend"/>
  </head>

  <results>

    <result>
      <binding name="x">
        <bnode>r2</bnode>
      </binding>
      <binding name="hpage">
        <uri>http://work.example.org/bob/</uri>
      </binding>
      <binding name="name">
        <literal xml:lang="en">Bob</literal>
      </binding>
      <binding name="age">
        <literal datatype="http://www.w3.org/2001/XMLSchema#integer">30</literal>
      </binding>
      <binding name="mbox">
        <uri>mailto:bob@work.example.org</uri>
      </binding>
    </result>

    ...
  </results>

</sparql>
```

Source: <http://www.w3.org/TR/rdf-sparql-XMLres/>

Implementations

- n Implementations

- n A list of related tools (clients, servers, parsers, ...)

- n <http://esw.w3.org/topic/SparqlImplementations>

- n In the lab session we will use Jena with ARQ

- n <http://jena.sourceforge.net/ARQ/>

Applications

- n SPARQL Endpoints:

- n DBLP: <http://dblp.l3s.de/d2r/>

- n DBPedia: <http://dbpedia.org/sparql>

- n Garlik.com

- n DataPatrol: reports on personal information online

- n Uses SPARQL to build the reports

- n Thousands of users

- n Billion triples in several repositories

- n Reports generated in few seconds

RDF Storage

Approaches to store RDF (I)

- n How do we store the RDF triples in a relational database?
- n Triple store
 - n Each RDF statement is stored as a single row in a three column 'statement' table
 - n Typically, a fourth column is added to indicate if the object is a literal or a URI

Approaches to store RDF (II)

- n Normalized triple store
 - n Triple store plus two additional tables:
 - n Resources table
 - n Literals table
 - n The statement table stores the subject, predicate and object as references to the values in the resources and literals tables
 - n Less space (each literal or URI stored only once)
 - n But complex queries (3-way join among the tables)
 - n Jena 1 uses this approach

Approaches to store RDF (III)

- n Denormalized triple store
 - n Hybrid of the standard triple store and the normalized triple store
 - n Three tables (statements, literals, resources)
 - n Short URIs/literals are directly stored in the statements table, long ones in the respective tables
 - n A threshold defines “long” vs “short” (Jena 2: 256)
 - n More space than normalized
 - n But in several cases the queries can be directly solved using only the information in the statements table
 - n Used by Jena 2

References (I)

- n SPARQL Query Language for RDF
 - n W3C Recommendation 15 January 2008
 - n <http://www.w3.org/TR/rdf-sparql-query/>
- n SPARQL Protocol for RDF
 - n W3C Recommendation 15 January 2008
 - n <http://www.w3.org/TR/rdf-sparql-protocol/>
- n SPARQL Query Results XML Format
 - n W3C Recommendation 15 January 2008
 - n <http://www.w3.org/TR/rdf-sparql-XMLres/>
- n Jena2 Database Interface - Database Layout
 - n <http://jena.sourceforge.net/DB/layout.html>

References (II)

- n Semantics and Complexity of SPARQL
 - n ISWC 2006: J. Perez, M. Arenas, C. Gutierrez
 - n <http://arxiv.org/abs/cs.DB/0605124>
- n SPARQL - Where are we? Current state, theory and practice
 - n ESWC 2007 Tutorial
 - n <http://axel.deri.ie/~axepol/sparqltutorial/>
- n SPARQL Tutorial
 - n Jena @ sourceforge
 - n <http://jena.sourceforge.net/ARQ/Tutorial/>